
IBIS modeling of buffers in ADS5545/46/47/25/27 family 
 

This document describes the method used to model the I/O buffers in 
ADS5545/46/47/25/27 family of products through IBIS. ( Sections 1-5 describe the 
model creation methodology and section 6-7 illustrate spice decks for use in simulations. 
Section 8 describes the contents of the model file ) 
The buffers modeled are:  

1. Input buffer for external clock input 
2. Tri-state-CMOS output buffer at the output data pin 
3. Tri-state-CMOS output buffer at the output clock pin 
4. Tri-state- LVDS ( DDR ) output buffer at the output data and clock pins  

The CMOS and the LVDS output buffers are tri-state buffers so that only one of them is 
activated depending on what mode of output is chosen. For more information refer to the 
datasheet (http://focus.ti.com/lit/ds/symlink/ads5546.pdf)  
 
1. Tri-state CMOS buffer  
 
The structure of the CMOS buffer in ADS5545/46/47/25/27 family of products is shown 
in Figure 1.1. The structure can be directly mapped to the structure of a tri-state output 
buffer as specified by the IBIS standard. So its model shall consist of  

1. I-V characteristics of Power and Ground clamp as well as the pull-up and pull-
down circuits.  

2. V-t characteristics of output pin  
Voltage levels of 0 Volts and 1.8 Volts correspond to logic levels of ‘0’ and ‘1’ at HI and 
LO at the input of the buffer.  
 

 
 

Figure 1.1 Tri-state CMOS output buffer structure 
 
1.1 V-I tables for Pull-up and Power clamp  
The Circuit setup used to extract the pull-up and power clamp data is shown in Fig. 1.2  
Two sets of measurements are taken. 

1 Measurement (a) 
o Input is at HI 

o Enable pin is at LO to deactivate output 
o The voltage source VPIN is swept from -VDD to 2*VDD ( as VSS = 0 ), 

where VDD = 3.3V, 3.0V or 3.6V (for typ, min and max corners ). done 
by first giving a LO at the output enable pin  

o The current flowing through VPIN is tabulated 
 
 

2 Measurement (b)  
o Input is at HI 
o Enable pin is at HI to activate output 
o The voltage source VPIN is swept from -VDD to 2*VDD ( as VSS = 0 ), 

where VDD = 3.3V, 3.0V or 3.6V (for typ, min and max corners ). 
o The current flowing through VPIN is tabulated 

 
The current measurements in (a) ( for VPIN = VDD to 2VDD )  are the Power clamp 
data, as only clamp circuits are active. The current measurement in (b) is sum of clamp 
circuit current and pull-up circuit current. So (b) - (a) gives the pull-up data.  

 
Figure 1.2 Circuit setup to extract Pull-up and Power clamp data 

1.2 V-I tables for Pull-down and Ground clamp  
 
The Circuit setup used to extract the pull-down and ground clamp data is shown in Fig. 
1.3. Two sets of measurements are taken.  

1. Measurement (a) 
o Input is at LO 
o Enable pin is at LO to deactivate output 
o The voltage source VPIN is swept from -VDD to 2*VDD ( as VSS = 0 ), 

where VDD = 3.3V, 3.0V or 3.6V (for typ, min and max corners ). done 
by first giving a LO at the output enable pin  

o The current flowing through VPIN is tabulated 
2. Measurement (b)  

o Input is at LO 
o Enable pin is at HI to activate output 
o The voltage source VPIN is swept from -VDD to 2*VDD ( as VSS = 0 ), 

where VDD = 3.3V, 3.0V or 3.6V (for typ, min and max corners ). 
o The current flowing through VPIN is tabulated 

The current measurements in (a) ( for VPIN = -VDD to VDD )  are tabulated as the 
Ground clamp data, as only clamp circuits are active. The current measurement in (b) is 
sum of clamp circuit current and pull-down circuit current. So (b) - (a) gives the pull-
down data.  

 

 
 

Figure 1.3 Circuit setup to extract Pull-down and Ground clamp data 
 

1.3 Generating the V-t Data 
Four V-t tables are generated using the setup shown in Figure 1.4. A step signal is given 
at input at voltage waveform at output is tabulated 
 

 
 

Figure 1.4 Circuit setup to extract the four V-t tables 

1.4 Comparison of IBIS model with the actual circuit 
To check how well the IBIS models shall model the actual circuit, the following SPICE 
simulations were done. The circuit set up is shown in Figure 1.5.  In the first simulation 
the SPICE netlist of the buffer is used and in the second simulation the IBIS model of the 
buffer is used.  

 
Figure 1.5 Circuit set up used in simulations for validation of the model 

 
Simulations were done with different values of the delay for the transmission line. 
 
Figure 1.6 shows a comparison of the HSPICE simulation results obtained from the IBIS 
model with those from the actual circuit.  
 

 
 
 

Figure 1.6 Output waveforms of simulation with IBIS model Vs. full-transistor netlist 
 

As can be observed, there is a close match between the actual circuit and its IBIS model. 
 
 
 



 7 

 
 

2. LVDS BUFFER (tri-state ) 
 Since the LVDS buffer has differential inputs and differential outputs, some additional 
considerations are involved while modeling it through IBIS. 
2.1 LVDS buffer architecture 
The internal structure of the LVDS buffer in ADS5545/46/47/25/27 family of products is 
shown in Figure 2.1. For simplicity the enable pin has not been shown. HI and LO refer 
to the differential input to the buffer. Voltage levels of 0 Volts and 1.8 Volts correspond 
to logic levels of ‘0’ and ‘1’ at HI and LO. LOUTP and LOUTM are the differential 
output of the buffer. The buffer consists of two current sources (one NMOS and one 
PMOS) and a set of four switches to change the direction of the current flowing into the 
external load.  

 
Figure 2.1 Internal structure of LVDS buffer 

 
While the NMOS current source is a constant current source (its value set by one of the 
eight current mode settings), the PMOS current source is controlled through a feedback 
loop as shown in Figure 2.2. 

 
Figure 2.2 Controlling the PMOS current source 

The PMOS current source is controlled so as to maintain the output common mode 
voltage of the buffer equal to a reference voltage of 1.2 Volts.  
2.2 Extracting V-I tables 
The setup for the pulldown and ground clamp data is as shown in Figure 2.3. A 50 Ohm 
resistor connects LOUTM to the common mode voltage of VBG=1.2 Volts. The 
measurements used for extracting the pull-up, pull-down, power clamp and the ground 
clamp data are very much similar to the method described for CMOS tri-state buffer in 
section 1.1 .  

 
Figure 2.3 Circuit set up for extracting pull-down and ground clamp data 

Since the NMOS current source is a constant current source, the pull-down current is 
fairly a constant for an output voltage between 0.5 V to 1.9 V.  
 
The pull-up and power clamp data is generated by the method shown in Figure 2.4. Refer 
to section 1.2 for a description of the method used. In this case however, because of the 
P-side current source being a controlled current source, the pull-up current source varies 
even around an output voltage of 1.2 Volts. 

 
Figure 2.4 Circuit setup to extract pull-up and power clamp data 



 9 

2.3 Generating the V-t table 
The V-t tables are generated using the setup shown in Figure 2.5.  
 

 
Figure 2.5 Circuit set up used to extract V-t table 

 
 
2.4 Modeling the internal termination resistances 
There are 3 internal termination resistances between the 2 output pins of the lvds buffer 
that can be turned on or off individually. These resistances have been modeled in the 
IBIS model file as series switch model. This model consists of a V-I table in which the 
current flowing in the resistor is tabulated (at the 3 corners) for a range of voltage across 
it. There are 3 separate models for the 3 individual resistances  

1. series_res_166 
2. series_res_200 
3. series_res_333 

2.5 Comparison of IBIS model with the actual circuit 
The IBIS models were generated and compared with the actual circuit through SPICE 
simulations. Figure 2.6 shows the simulation environment for the actual LVDS buffer, 
while Figure 2.7 shows the identical environment used for the IBIS model. 
 

 
Figure 2.6 Circuit set up of the full transistor spice deck 

 
 

 
Figure 2.7 Circuit set up of spice deck using IBIS models 

 
Simulations were done with different values of the delay for the transmission line. 
 
Figure 2.8 shows a comparison of the SPICE simulation results obtained from the IBIS 
model with those from the actual circuit. The delay of the transmission line was set to 
2.0ns. There are 2 sets of waveforms 

1. LOUTP and LOUTM : waveforms at the buffer output 
2. LOUTP_RX and LOUTM_RX: Waveform at the receiver end. 

The test circuits shown in Figure 2.6 / 2.7 have the lvds buffer without the internal load 
terminations. When internal load terminations are used there is a very small shift in the 
common mode of the output wave forms (~ 0.05 V ). But the shape of the waveforms as 
well as the differential signal shall well correlate with SPICE results. This is due to the 
inherent inability of IBIS modeling to model feedback (As mentioned in section 2.1 the 
common mode is set using a feedback loop ).     

Figure 2.8 

 
 
As can be observed, there is a close match between the actual circuit and its IBIS model. 

3. Input buffer for clock 
 
The structure of the input buffer in ADS5545/46/47/25/27 family of products is shown in 
Figure 3.1. The inputs to the buffer are differential and internally it consists of clamping 
circuits and pull-up/pull-down circuits (as in any input buffer). In addition to them, there 
is a termination consisting of a capacitance C and 2 equal resistances R1 and R2 from 
each input to a common mode voltage of 1.5V.  

 
Figure 3.1 Structure of the input buffer in ADS5545/46/47/25/27 family 

 
As it is an input buffer the pull-up/pull-down circuits shall not be characterized in the 
IBIS model. The differential input buffer plus R1 and R2 is modeled as combination of 2 
single-input buffers and termination capacitance C is modeled as a series element. So the 
IBIS model file consists of: 

1. A model for characterizing clamp circuits + termination resistance. Model type 
used is Terminator 

2. A model for the termination capacitance. Model type used is Series 
 
The input file to the simulator shall consist of 2 instances of the 1st model to represent the 
differential input and 1 instance of 2nd model to represent the capacitive termination 
between the differential inputs 
 
3.1 Model for characterizing clamp circuits and termination resistance 
 
The clamp circuit together with termination resistance (R1 or R2) shall be modeled using 
buffer type Terminator. The model name in the ibis file is ‘input’.  
 
3.1.1 Extracting power clamp and ground clamp data 
 
The setup for the CLAMP data extraction is as shown in Figure 3.2.  

Figure 3.2 Circuit setup used to extract V-I tables  

 
3.1.2 Termination resistance 
 
As shown in Figure 3.1 the termination resistance is connected between input pin and a 
constant voltage of 1.5V. The Terminator buffer model has keywords [Rpower] and 
[Rgnd] to specify resistance between the pin to supply voltage and ground respectively. 
So we split the termination resistance into 2 parts ; one each to the supply and to the 
ground. The values of Rpower and Rgnd are obtained by solving 
VDD*Rpower / ( Rpower + Rgnd ) = 1.5 
Rpower*Rgnd / ( Rpower + Rgnd ) = R1 ( = R2 ) 
This calculation is done for ‘typ’, ‘min’ as well as ‘max’ values of R1(=R2) to get the 
Rpower and Rgnd at the 3 corners. 
 
From the above discussion, it follows that the ‘Terminator’ model can be represented 
diagrammatically as in Figure 3.3 

 
Figure 3.3 The equivalent terminator model of the termination resistance  

3.2 Termination Capacitance 
 
The termination capacitance ( C ) between the pins of the differential input is captured in 
the IBIS model as ‘Series’ buffer type. The [C Series] keyword is used to specify its 
value at the 3 corners. 
 
3.3 Comparison of IBIS model with the actual circuit 
The IBIS models were generated and compared with the actual circuit through SPICE 
simulations. Figure 3.4 shows the simulation environment. 

 
Figure 3.4 Circuit set up used in simulations to validate the IBIS model 

 
Two simulations are performed. The first one uses the full transistor spice netlist of the 
input buffer circuit. The second simulation uses the IBIS model representation of the 
buffer as shown in Figure 3.5. A sample spice deck for this configuration is given in 
section 7.3 of the document. 

 
Figure 3.5 The “clock input buffer” block in Fig 3.4 was replaced with this setup of  

model instances for simulation with IBIS model 
 
 
Simulations were done with different values of the delay for the transmission line. 



 15 

 
Figure 3.6 shows a comparison of the SPICE simulation results obtained from the IBIS 
model with those from the actual circuit. The transmission line used has Z0 = 50 ohms 
and delay = 1ns.  
 

 
 

Figure 3.6 Wave forms of simulation with IBIS model Vs full transistor netlist  
 
 
There is a close match between the output waveforms of simulation using actual circuit 
netlist and its IBIS model. 

4. MODEL SUMMARY 
 
Component name -  ADS554X * 
Model types -  Tristate, Terminator, series switch, series cap 
Filename -  ads554x.ibs * 
 
Modeling conditions: 
 
Condition typ/ min/ max 
VDD 3.3 V/ 3.0 V/ 3.6 V 
VCM (Common mode 
voltage) 

1.5V 

Junction temperature (Tj) 27/ 125/ -40 (degree C) 
Process setting nominal/ weak/ strong 
 

Package characteristics : 

Characteristics typ/ min/ max 
R_pkg: 0.438 / 0.327 / 0.55 (ohms) 
L_pkg 1.442nH  /1.022nH / 1.87nH 
C_pkg 0.13pF /0.084pF / 0.201pF 
 
5. Quality Checks :  
 
The created IBIS models are verified as follows: 

1. Visual check of each characteristics using the s2iplt tool 
2. Syntax check by IBISCHK4 
3. Behavior of IBIS models and actual circuit compared using SPICE 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
* Even though the component name is specified as ADS554X in the IBIS model, the 
same model can be used for all of the ADS5545/46/47/25/27 family of products.

6. Sample instantiation of the IBIS models in HSPICE: 
 
6.1 CMOS DATA OUTPUT BUFFER 
 
BOUTPUTP PU 0 OUTP INP EN_CMOS 
+FILE = 'ads554x.ibs' MODEL = 'cmos_obuf_data' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
*ENABLE SIGNAL FOR THE BUFFER 
VEN_CMOS EN_CMOS 0 3.3  
 
6.2 CMOS CLOCK OUTPUT BUFFER 
 
BOUTPUTP PU 0 OUTP INP EN_CMOS 
+FILE = 'ads554x.ibs' MODEL = ‘cmos_obuf_clk’ 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
*ENABLE SIGNAL FOR THE BUFFER 
VEN_CMOS EN_CMOS 0 3.3  
 
6.3 LVDS OUTPUT BUFFER (SAME FOR BOTH DATA AND CLOCK) ALONG 
WITH A TERMINATION RESISTANCE  
 
BOUTPUTP PU 0 LOUTP INP EN_LVDS 
+FILE = 'ads554x.ibs' MODEL = 'lvds_obuf_1.95' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
BOUTPUTM PU 0 LOUTM INM EN_LVDS 
+FILE = 'ads554x.ibs' MODEL = 'lvds_obuf_1.95' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 

VEN_LVDS EN_LVDS 0 3.3 
 
*INTERNAL LOAD TERMINATION OF 166 OHM 
BSERIES166 LOUTP LOUTM  
+FILE = 'ads554x.ibs' MODEL = 'series_res_166' 
+TYP = TYP  
+SS_STATE = ON 
+BUFFER = SERIES_SWITCH 
+INTERPOL = 1 
 
6.4 INPUT BUFFER FOR CLOCK ALONG WITH TERMINATION 
CAPACITANCE 
 
VAVDD AVDD 0 3.3 
VAVSS AVSS 0 0 
 
BCLKINP  AVDD AVSS CLKP   
+FILE ='ads554x.ibs' MODEL = 'input' 
+TYP = TYP POWER=OFF 
+BUFFER = TERMINATOR 
+INTERPOL = 1 
 
BCLKINM AVDD AVSS CLKM  
+FILE ='ads554x.ibs' MODEL = 'input' 
+TYP = TYP POWER=OFF 
+BUFFER = TERMINATOR 
+INTERPOL = 1 
 
*The following line may also be replaced with a capacitance  
*of value as specified against the [C Series] keyword in IBIS model 
BSERIES_CAP CLKP CLKM 
+FILE ='ads554x.ibs' MODEL = 'series_cap' 
+TYP = MIN 
+BUFFER = SERIES 
+INTERPOL = 1 

7. Sample SPICE deck for simulation of the buffer in HSPICE 
 
7.1 CMOS DATA OUTPUT BUFFER 
 
 
.TEMP 27 
* INPUT TO THE BUFFER 
VINP INP 0 PULSE (1.8 0 0N 0.2N 0.2N 5.6N 11.6N) 
 
********************** 
*     Buffer         * 
********************** 
BOUTPUTP PU 0 OUTP INP EN_CMOS 
+FILE = 'ads554x.ibs' MODEL = 'cmos_obuf_data' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
*ENABLE SIGNAL FOR THE BUFFER 
VEN_CMOS EN_CMOS 0 3.3  
 
* PIN INDUCTANCE 
LOUTP OUTP_EXT OUTP 1.87N 
* PIN CAPACITANCE 
COUTP OUTP_EXT 0 0.2PF 
 
* TRANSMISSION LINE LOAD 
TDATALINKP OUTP_EXT 0 OUTP_RX 0 Z0=50 TD=2N 
 
* LVDS TERMINATION AT RECEIVER END 
RLOAD OUTP_RX 0 50 
 
* CAPACITANCE AT RECEIVER END 
COUTP_RX OUTP_RX 0 2PF 
 
.OPTION POST 
.OPTION DELMAX=1PS 
.TRAN 10PS 50NS 
.PLOT TRAN V(OUTP)  
+ V(OUTP_EXT)  
+ V(OUTP_RX) 
.END 

7.2 CMOS CLOCK OUTPUT BUFFER 
 
 
.TEMP 27 
* INPUT TO THE BUFFER 
VINP INP 0 PULSE (1.8 0 0N 0.2N 0.2N 5.6N 11.6N) 
 
********************** 
*     Buffer         * 
********************** 
BOUTPUTP PU 0 OUTP INP EN_CMOS 
+FILE = 'ads554x.ibs' MODEL = ‘cmos_obuf_clk’ 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
*ENABLE SIGNAL FOR THE BUFFER 
VEN_CMOS EN_CMOS 0 3.3  
 
* PIN INDUCTANCE 
LOUTP OUTP_EXT OUTP 1.87N 
* PIN CAPACITANCE 
COUTP OUTP_EXT 0 0.2PF 
 
* TRANSMISSION LINE LOAD 
TDATALINKP OUTP_EXT 0 OUTP_RX 0 Z0=50 TD=2N 
 
* LVDS TERMINATION AT RECEIVER END 
RLOAD OUTP_RX 0 50 
 
* CAPACITANCE AT RECEIVER END 
COUTP_RX OUTP_RX 0 2PF 
 
.OPTION POST 
.OPTION DELMAX=1PS 
.TRAN 10PS 50NS 
.PLOT TRAN V(OUTP)  
+ V(OUTP_EXT)  
+ V(OUTP_RX) 
.END 

7.3 LVDS OUTPUT BUFFER (FOR DATA and CLOCK) ALONG WITH A 
TERMINATION RESISTANCE  
 
.TEMP 27 
 
* INPUT TO THE BUFFER 
VINM INM 0 PULSE (0 1.8 0N 0.2N 0.2N 2.7N 5.8N) 
VINP INP 0 PULSE (1.8 0 0N 0.2N 0.2N 2.7N 5.8N) 
 
********************** 
*     Buffer         * 
********************** 
BOUTPUTP PU 0 LOUTP INP EN_LVDS 
+FILE = 'ads554x.ibs' MODEL = 'lvds_obuf_1.95' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
BOUTPUTM PU 0 LOUTM INM EN_LVDS 
+FILE = 'ads554x.ibs' MODEL = 'lvds_obuf_1.95' 
+TYP = TYP POWER=ON 
+BUFFER = THREE_STATE 
+RAMP_RWF = 1 
+RAMP_FWF = 1 
+INTERPOL = 1 
 
VEN_LVDS EN_LVDS 0 3.3 
 
*INTERNAL LOAD TERMINATION OF 166 OHM 
BSERIES166 LOUTP LOUTM  
+FILE = 'ads554x.ibs' MODEL = 'series_res_166' 
+TYP = TYP  
+SS_STATE = ON 
+BUFFER = SERIES_SWITCH 
+INTERPOL = 1 
 
* PIN INDUCTANCE 
LOUTP LOUTP_EXT LOUTP 1.87N 
LOUTM LOUTM_EXT LOUTM 1.87N 
** PIN CAPACITANCE 
COUTP LOUTP_EXT 0 0.2PF 
COUTM LOUTM_EXT 0 0.2PF 
 
** TRANSMISSION LINE LOAD 

TDATALINKP LOUTP 0 LOUTP_RX 0 Z0=50 TD=2N 
TDATALINKM LOUTM 0 LOUTM_RX 0 Z0=50 TD=2N 
 
** LVDS TERMINATION AT RECEIVER END 
RLOAD LOUTP_RX LOUTM_RX 100 
* 
** CAPACITANCE AT RECEIVER END 
COUTP_RX LOUTP_RX 0 2PF 
COUTM_RX LOUTM_RX 0 2PF 
 
.OPTION POST 
.TRAN 10PS 100NS 
.PLOT TRAN V(LOUTP) V(LOUTM)  
+ V(LOUTP_EXT) V(LOUTM_EXT)  
+ V(LOUTP_RX) V(LOUTM_RX)  
+ V(LOUTP_RX,LOUTM_RX) 
 
.END 
 
 
7.4 INPUT BUFFER FOR CLOCK ALONG WITH TERMINATION 
CAPACITANCE 
 
.TEMP 27 
VAVDD AVDD 0 3.3 
VAVSS AVSS 0 0 
 
BCLKINP  AVDD AVSS CLKP   
+FILE ='ads554x.ibs' MODEL = 'input' 
+TYP = TYP POWER=OFF 
+BUFFER = TERMINATOR 
+INTERPOL = 1 
 
BCLKINM AVDD AVSS CLKM  
+FILE ='ads554x.ibs' MODEL = 'input' 
+TYP = TYP POWER=OFF 
+BUFFER = TERMINATOR 
+INTERPOL = 1 
 
*The following line may also be replaced with a capacitance  
*of value as specified against the [C Series] keyword in IBIS model 
BSERIES_CAP CLKP CLKM 
+FILE ='ads554x.ibs' MODEL = 'series_cap' 
+TYP = MIN 
+BUFFER = SERIES 
+INTERPOL = 1 

VCLKP LVPECLP 0 DC 0  
+PWL 0 1.1 1n 1.1 1.3n 1.9 3.9n 1.9 4.2n 1.1 6.8n 1.1 R=1n 
 
VCLKM LVPECLM 0 DC 0  
+PWL 0 1.9 1n 1.9 1.3n 1.1 3.9n 1.1 4.2n 1.9 6.8n 1.9 R=1n 
 
T1 LVPECLP 0 CLKP1 0 Z0=50 TD=1N 
T2 LVPECLM 0 CLKM1 0 Z0=50 TD=1N 
 
R1P AVDD CLKP1 130 
R2P AVSS CLKP1 82 
CC1P CLKP1 CLKPX 5pF 
 
R1M AVDD CLKM1 130 
R2M AVSS CLKM1 82 
CC1M CLKM1 CLKMX 5pF 
 
C1 CLKPX 0 0.084pF 
L1 CLKPX CLKP 1.022n 
 
C2 CLKMX 0 0.084pF 
L2 CLKMX CLKM 1.022n 
 
.TRAN 0.01n 1000n 
.OPTIONS POST 
.plot tran V(CLKP CLKM) 
.END 

8. A brief description of contents of the model file : ads554x.ibs 
 
The list of pins of the device and corresponding model names for them are tabulated 
under the key word [Pin]. The names clockoutp, clockoutm and dataoutput represent 
model selectors. i.e. The pins corresponding to them have multiple modes of operation 
and there are separate models for each mode of operation. The various models 
corresponding to each model selector is specified under the keyword [Model Selector].  
 
8.1 DATA OUTPUT BUFFER  
There are 2 modes of operation: lvds mode and cmos buffer mode. The lvds mode itself 
has 8 modes of operation which correspond to different drive strengths for the output 
transistors. Hence there are a total of 8 + 1 = 9 models specified under [Model Selector] 
dataoutput. ( The circuit for cmos buffers are different for clock and data because clock 
rate is double of data rate; but the lvds buffer is identical as in that mode it is DDR. 
Hence lvds buffer names are same for both whereas cmos buffer model name is different  
). Also the lvds buffers have 3 termination resistances b/w the 2 output pins which can be 
turned on or off. They have been separately modeled (see section 8.4 below) 
8.2 CLOCK OUTPUT BUFFER  
There are 2 modes of operation: lvds mode and cmos buffer mode. The lvds mode itself 
has 8 modes of operation which correspond to different drive strengths for the output 
transistors. In the cmos mode, one of the 2 clock output pins is inactive and in lvds mode 
both are active. Hence for 1st pin there are total 8 + 1 = 9 models specified under [Model 
Selector]clockoutp. And for 2nd pin there are 8 models specified under [Model 
Selector]clockoutm. Also the lvds buffers have 3 termination resistances b/w the 2 output 
pins which can be turned on or off. They have been separately modeled (see section 8.4 
below) 
8.3 Clock input buffer 
The clock input buffer has model name “input”. It also has a terminating capacitance 
between the 2 input pins which has been separately specified as model series_cap (see 
section 8.4 ) 
8.4 Termination resistances for lvds buffers and capacitance for input buffer 
The lvds buffers have 3 termination resistances between the 2 output pins which can be 
turned on or off. Their model name is series_res_<value>. Also there is a termination 
capacitance between the input pins of clock buffer. It is captured under the model name 
series_cap.  
The keyword [Series Pin Mapping] lists the pairs of pins across which these termination 
elements are present 
 


